Appolonian sphere packing -- Derivation of "special" coordinates and inversion formula
Author: Ronny Peikert

```
> with(linalg):
```

Abbreviations for square roots

```
> q2 := 2^(1/2):  q3 := 3^(1/2):  q6 := 6^(1/2):
```

Simplification procedure

```
> mysimp := proc(e)
      expand(radsimp(simplify(e), ratdenom));
  end:
```

Inversion spheres, expressed in Cartesian coordinates plus radius, (x,y,z,r)

```
> inv1c := [0, 0, 0, q3 - q2];
  inv2c := [q3, -q3, -q3, 2*q2];
  inv3c := [-q3,  q3, -q3, 2*q2];
  inv4c := [-q3, -q3, q3, 2*q2];
  inv5c := [q3, q3, q3, 2*q2];
```

$$inv1c := \left[0, 0, 0, \sqrt{3} - \sqrt{2}\right]$$

$$inv2c := \left[\sqrt{3}, -\sqrt{3}, -\sqrt{3}, 2\sqrt{2}\right]$$

$$inv3c := \left[-\sqrt{3}, \sqrt{3}, -\sqrt{3}, 2\sqrt{2}\right]$$

$$inv4c := \left[-\sqrt{3}, -\sqrt{3}, \sqrt{3}, 2\sqrt{2}\right]$$

$$inv5c := \left[\sqrt{3}, \sqrt{3}, \sqrt{3}, 2\sqrt{2}\right] \quad (1)$$

Initial spheres (generators), expressed in Cartesian coordinates plus radius, (x,y,z,r)

```
> init1c := [0, 0, 0, -1];
  init2c := [q2-q3, q3-q2, q3-q2, q6-2 ];
  init3c := [q3-q2, q2-q3, q3-q2, q6-2 ];
  init4c := [q3-q2, q3-q2, q2-q3, q6-2 ];
  init5c := [q2-q3, q2-q3, q2-q3, q6-2 ];
```

$$init1c := [0, 0, 0, -1]$$

$$init2c := \left[-\sqrt{3} + \sqrt{2}, \sqrt{3} - \sqrt{2}, \sqrt{3} - \sqrt{2}, -2 + \sqrt{6}\right]$$

$$init3c := \left[\sqrt{3} - \sqrt{2}, -\sqrt{3} + \sqrt{2}, \sqrt{3} - \sqrt{2}, -2 + \sqrt{6}\right]$$

$$init4c := \left[\sqrt{3} - \sqrt{2}, \sqrt{3} - \sqrt{2}, -\sqrt{3} + \sqrt{2}, -2 + \sqrt{6}\right]$$

$$init5c := \left[-\sqrt{3} + \sqrt{2}, -\sqrt{3} + \sqrt{2}, -\sqrt{3} + \sqrt{2}, -2 + \sqrt{6}\right] \quad (2)$$

Transformation from (x,y,z,r) to inversive coordinates (a,b,c,d,e) -- taken from Boyd's (?) papers

```
> cartToInv := proc(xyzr)
      local x, y, z, r, a, b, c, d, e;
      x := xyzr[1]; y := xyzr[2]; z := xyzr[3]; r := xyzr[4];
      a := x/r;
      b := y/r;
      c := z/r;
```

```
      d := (x^2 + y^2 + z^2 - r^2 - 1) / (2*r);
      e := (x^2 + y^2 + z^2 - r^2 + 1) / (2*r);
      map(mysimp, [a, b, c, d, e]);
  end:
```

Back transformation

```
> invToCart := proc(abcde)
      local a, b, c, d, e, x, y, z, r;
      a := abcde[1];
      b := abcde[2];
      c := abcde[3];
      d := abcde[4];
      e := abcde[5];
      r := 1 / (e - d);
      x := a * r;
      y := b * r;
      z := c * r;
      map(mysimp, [x, y, z, r]);
  end:
```

Transformation of initial spheres

```
> init1i := cartToInv(init1c);
  init2i := cartToInv(init2c);
  init3i := cartToInv(init3c);
  init4i := cartToInv(init4c);
  init5i := cartToInv(init5c);
```

$$init1i := [0, 0, 0, 1, 0]$$

$$init2i := \left[-\frac{1}{2}\sqrt{2}, \frac{1}{2}\sqrt{2}, \frac{1}{2}\sqrt{2}, -1, \frac{1}{2}\sqrt{3}\sqrt{2}\right]$$

$$init3i := \left[\frac{1}{2}\sqrt{2}, -\frac{1}{2}\sqrt{2}, \frac{1}{2}\sqrt{2}, -1, \frac{1}{2}\sqrt{3}\sqrt{2}\right]$$

$$init4i := \left[\frac{1}{2}\sqrt{2}, \frac{1}{2}\sqrt{2}, -\frac{1}{2}\sqrt{2}, -1, \frac{1}{2}\sqrt{3}\sqrt{2}\right]$$

$$init5i := \left[-\frac{1}{2}\sqrt{2}, -\frac{1}{2}\sqrt{2}, -\frac{1}{2}\sqrt{2}, -1, \frac{1}{2}\sqrt{3}\sqrt{2}\right] \qquad \textbf{(3)}$$

Transformation from inversive coordinates (a,b,c,d,e) to "special" coordinates (A,B,C,D,E). The matrix T is chosen such that it maps the 5 basis vectors to the vectors init1i, ..., init5i

```
> T := transpose(matrix([init1i, init2i, init3i, init4i, init5i]))
  ;
```

$$T := \begin{bmatrix} 0 & -\frac{1}{2}\sqrt{2} & \frac{1}{2}\sqrt{2} & \frac{1}{2}\sqrt{2} & -\frac{1}{2}\sqrt{2} \\ 0 & \frac{1}{2}\sqrt{2} & -\frac{1}{2}\sqrt{2} & \frac{1}{2}\sqrt{2} & -\frac{1}{2}\sqrt{2} \\ 0 & \frac{1}{2}\sqrt{2} & \frac{1}{2}\sqrt{2} & -\frac{1}{2}\sqrt{2} & -\frac{1}{2}\sqrt{2} \\ 1 & -1 & -1 & -1 & -1 \\ 0 & \frac{1}{2}\sqrt{3}\sqrt{2} & \frac{1}{2}\sqrt{3}\sqrt{2} & \frac{1}{2}\sqrt{3}\sqrt{2} & \frac{1}{2}\sqrt{3}\sqrt{2} \end{bmatrix} \tag{4}$$

```
> TI := inverse(T):
```

```
> invToSpec := proc(abcde)
      map(mysimp, evalm(TI &* abcde));
  end:
```

Back transformation

```
> specToInv := proc(ABCDE)
      map(mysimp, evalm(T &* ABCDE));
  end:
```

Compound transformation from Cartesian to special coordinates

```
> cartToSpec := proc(xyzr)
      invToSpec(cartToInv(xyzr));
  end:
```

Back transformation

```
> specToCart := proc(ABCDE)
      invToCart(specToInv(ABCDE));
  end:
```

Inversion of a sphere p at a sphere q, in inversive coordinates.

```
> iproduct := proc(u, v)
      u[1]*v[1] +
      u[2]*v[2] +
      u[3]*v[3] +
      u[4]*v[4] -
      u[5]*v[5];
  end:
```

```
> reflectAt := proc(p, q)
      local r:
      r := p - 2*iproduct(p,q)*q;
  end:
```

Inversion of the sphere p at the sphere q, in special coordinates

```
> reflectAt_s := proc(p, q)
```

```
      local pi, qi, ri, r;
      pi := specToInv(p);
      qi := specToInv(q);
      ri := reflectAt(pi,qi);
      r := invToSpec(ri);
      map(mysimp, r);
  end:
```

Inversion spheres, in special coordinates

```
> inv1s := cartToSpec(inv1c);
  inv2s := cartToSpec(inv2c);
  inv3s := cartToSpec(inv3c);
  inv4s := cartToSpec(inv4c);
  inv5s := cartToSpec(inv5c);
```

$$inv1s := \left[-\frac{1}{3}\sqrt{3} \quad \frac{1}{6}\sqrt{3} \quad \frac{1}{6}\sqrt{3} \quad \frac{1}{6}\sqrt{3} \quad \frac{1}{6}\sqrt{3} \right]$$

$$inv2s := \left[\frac{1}{6}\sqrt{3} \quad -\frac{1}{3}\sqrt{3} \quad \frac{1}{6}\sqrt{3} \quad \frac{1}{6}\sqrt{3} \quad \frac{1}{6}\sqrt{3} \right]$$

$$inv3s := \left[\frac{1}{6}\sqrt{3} \quad \frac{1}{6}\sqrt{3} \quad -\frac{1}{3}\sqrt{3} \quad \frac{1}{6}\sqrt{3} \quad \frac{1}{6}\sqrt{3} \right]$$

$$inv4s := \left[\frac{1}{6}\sqrt{3} \quad \frac{1}{6}\sqrt{3} \quad \frac{1}{6}\sqrt{3} \quad -\frac{1}{3}\sqrt{3} \quad \frac{1}{6}\sqrt{3} \right]$$

$$inv5s := \left[\frac{1}{6}\sqrt{3} \quad \frac{1}{6}\sqrt{3} \quad \frac{1}{6}\sqrt{3} \quad \frac{1}{6}\sqrt{3} \quad -\frac{1}{3}\sqrt{3} \right] \qquad \textbf{(5)}$$

Initial spheres (generators), in special coordinates (= basis vectors by construction)

```
> init1s := cartToSpec(init1c);
  init2s := cartToSpec(init2c);
  init3s := cartToSpec(init3c);
  init4s := cartToSpec(init4c);
  init5s := cartToSpec(init5c);
```

$$init1s := \left[1 \quad 0 \quad 0 \quad 0 \quad 0 \right]$$

$$init2s := \left[0 \quad 1 \quad 0 \quad 0 \quad 0 \right]$$

$$init3s := \left[0 \quad 0 \quad 1 \quad 0 \quad 0 \right]$$

$$init4s := \left[0 \quad 0 \quad 0 \quad 1 \quad 0 \right]$$

$$init5s := \left[0 \quad 0 \quad 0 \quad 0 \quad 1 \right] \qquad \textbf{(6)}$$

Spherical inversion (reflection at the 5 inversion spheres) of a general sphere, in special coordinates

```
> reflectAt_s([A,B,C,D,E], inv1s);
  reflectAt_s([A,B,C,D,E], inv2s);
  reflectAt_s([A,B,C,D,E], inv3s);
  reflectAt_s([A,B,C,D,E], inv4s);
  reflectAt_s([A,B,C,D,E], inv5s);
```

$$[\,-A \quad B+A \quad C+A \quad D+A \quad E+A\,]$$

$$[\,B+A \quad -B \quad B+C \quad B+D \quad B+E\,]$$

$$[\,C+A \quad B+C \quad -C \quad C+D \quad C+E\,]$$

$$[\,D+A \quad B+D \quad C+D \quad -D \quad D+E\,]$$

$$[\,E+A \quad B+E \quad C+E \quad D+E \quad -E\,] \qquad \textbf{(7)}$$

Testing -- First generation of spheres (= initial sphere #i reflected at inversion circle #i), in special coordinates

```
> sphere11s := reflectAt_s(init1s, inv1s);
  sphere22s := reflectAt_s(init2s, inv2s);
  sphere33s := reflectAt_s(init3s, inv3s);
  sphere44s := reflectAt_s(init4s, inv4s);
  sphere55s := reflectAt_s(init5s, inv5s);
```

$$sphere11s := [\,-1 \quad 1 \quad 1 \quad 1 \quad 1\,]$$

$$sphere22s := [\,1 \quad -1 \quad 1 \quad 1 \quad 1\,]$$

$$sphere33s := [\,1 \quad 1 \quad -1 \quad 1 \quad 1\,]$$

$$sphere44s := [\,1 \quad 1 \quad 1 \quad -1 \quad 1\,]$$

$$sphere55s := [\,1 \quad 1 \quad 1 \quad 1 \quad -1\,] \qquad \textbf{(8)}$$

... which is in Cartesian coordinates:

```
> evalf(specToCart(sphere11s));
  evalf(specToCart(sphere22s));
  evalf(specToCart(sphere33s));
  evalf(specToCart(sphere44s));
  evalf(specToCart(sphere55s));
```

$$[0., 0., 0., 0.101020514]$$

$$[0.4099776108, -0.4099776108, -0.4099776108, 0.2898979486]$$

$$[-0.4099776108, 0.4099776108, -0.4099776108, 0.2898979486]$$

$$[-0.4099776108, -0.4099776108, 0.4099776108, 0.2898979486]$$

$$[0.4099776108, 0.4099776108, 0.4099776108, 0.2898979486] \qquad \textbf{(9)}$$

Second generation of spheres, in special coordinates

```
> sphere112s := reflectAt_s(sphere11s, inv2s);
  sphere113s := reflectAt_s(sphere11s, inv3s);
  sphere114s := reflectAt_s(sphere11s, inv4s);
  sphere115s := reflectAt_s(sphere11s, inv5s);
  sphere221s := reflectAt_s(sphere22s, inv1s);
  sphere223s := reflectAt_s(sphere22s, inv3s);
```

$$sphere112s := \begin{bmatrix} 0 & -1 & 2 & 2 & 2 \end{bmatrix}$$

$$sphere113s := \begin{bmatrix} 0 & 2 & -1 & 2 & 2 \end{bmatrix}$$

$$sphere114s := \begin{bmatrix} 0 & 2 & 2 & -1 & 2 \end{bmatrix}$$

$$sphere115s := \begin{bmatrix} 0 & 2 & 2 & 2 & -1 \end{bmatrix}$$

$$sphere221s := \begin{bmatrix} -1 & 0 & 2 & 2 & 2 \end{bmatrix}$$

$$sphere223s := \begin{bmatrix} 2 & 0 & -1 & 2 & 2 \end{bmatrix} \quad \textbf{(10)}$$

```
>
```

etc.